



An analytical benchmark and a *Mathematica* program for MD codes: Testing LAMMPS on the 2nd generation Brenner potential[☆]



Antonino Favata^{a,*}, Andrea Micheletti^b, Seunghwa Ryu^c, Nicola M. Pugno^{d,e,f}

^a Department of Structural and Geotechnical Engineering, Sapienza University of Rome, Italy

^b Dipartimento di Ingegneria Civile e Ingegneria Informatica, University of Rome TorVergata, Italy

^c Department of Mechanical Engineering, Korea Advanced Institute of Science and Technology (KAIST), Daejeon 34141, Republic of Korea

^d Laboratory of Bioinspired and Graphene Nanomechanics, Department of Civil, Environmental and Mechanical Engineering, University of Trento, Italy

^e Center for Materials and Microsystems, Fondazione Bruno Kessler, Trento, Italy

^f School of Engineering and Materials Science, Queen Mary University of London, UK

ARTICLE INFO

Article history:

Received 4 September 2015

Received in revised form

11 June 2016

Accepted 13 June 2016

Available online 1 July 2016

Keywords:

REBO potentials

2nd generation Brenner potential

LAMMPS

Benchmark

Carbon nanotubes

ABSTRACT

An analytical benchmark and a simple consistent *Mathematica* program are proposed for graphene and carbon nanotubes, that may serve to test any molecular dynamics code implemented with REBO potentials. By exploiting the benchmark, we checked results produced by LAMMPS (Large-scale Atomic/Molecular Massively Parallel Simulator) when adopting the second generation Brenner potential, we made evident that this code in its current implementation produces results which are offset from those of the benchmark by a significant amount, and provide evidence of the reason.

Program summary

Program title: MDBenchmarks

Catalogue identifier: AFAS_v1_0

Program summary URL: http://cpc.cs.qub.ac.uk/summaries/AFAS_v1_0.html

Program obtainable from: CPC Program Library, Queen's University, Belfast, N. Ireland

Licensing provisions: GNU GPL v3

No. of lines in distributed program, including test data, etc.: 22854

No. of bytes in distributed program, including test data, etc.: 369171

Distribution format: tar.gz

Programming language: Mathematica 9.

Computer: Any PC.

Operating system: Any which supports Mathematica; tested under OS Yosemite.

RAM: <5 gigabytes

Classification: 7.7, 16.1, 16.13.

Nature of problem: Testing commercial or open-source molecular dynamics codes implementing off-the-shelf REBO potentials on an analytical benchmark.

Solution method: Analytical equilibrium conditions for achiral carbon nanotubes are implemented and solved, delivering benchmark values for the corresponding natural radius and cohesive energy; material properties (Young's modulus and Poisson coefficient) are also computed.

Running time: Instantaneous, or a few seconds, depending on computer hardware

© 2016 Elsevier B.V. All rights reserved.

[☆] This paper and its associated computer program are available via the Computer Physics Communication homepage on ScienceDirect (<http://www.sciencedirect.com/science/journal/00104655>).

* Corresponding author.

E-mail address: antonino.favata@uniroma1.it (A. Favata).

<http://dx.doi.org/10.1016/j.cpc.2016.06.005>

0010-4655/© 2016 Elsevier B.V. All rights reserved.

1. Introduction

Molecular dynamics (MD) simulations are nowadays more and more popular in scientific applications, especially in those fields of material science involving nanotechnology and advanced material design. On one side, there are advantages in the speed and accuracy of the simulations, with the model of the potential for atomic interactions being optimized to reproduce either experimental values or quantities estimated by first principles calculations (considered, as a matter of facts, just like experimental results). On the other side, it is more and more frequent to use commercial or open-source codes implementing off-the-shelf potential models, and use them as a black box, without having a precise feeling with the code itself. One of the most used simulator is LAMMPS (Large-scale Atomic/Molecular Massively Parallel Simulator), able to implement several interatomic potentials. By using an analytical discrete mechanical model, we present a benchmark for the equilibrium problem of graphene and carbon nanotubes, which can be applied to any kind of REBO (reactive empirical bond-order) potential. The analytical condition proposed produces results in complete agreement with First Principles, Density Functional Theory and Monte Carlo simulations. With the aid of this benchmark, we show that LAMMPS code, when implemented with the second generation Brenner potential, produces results which are offset from those of the benchmark by a significant amount, and provide evidence of the reason. The analytical formulation is implemented in a *Mathematica* program, intended to provide a set of easy-to-get benchmark solutions; combination of symbolic manipulation and numerical routines make the program easy to be adapted to any REBO potential, providing a general tool for testing MD codes.

2. An analytical discrete model for equilibrium configurations of FGSS and CNTs

The benchmark solution we propose has been developed within the context of carbon macromolecules, such as Flat Graphene Strips (FGSS) or Carbon Nanotubes (CNTs). When regarded from the point of view of MD, such aggregates are modeled as sets of mass points, whose configuration is described by the Cartesian coordinates of each point with respect to a chosen reference frame; each point is then interacting with the others – at least with the closest ones – and the interaction is captured by a suitable empirical potential, whose shape and parameters are fitted with a set of selected experiments and *ab initio* calculations. The last generation potentials usually take into account multi-particle interactions, up to the third nearest neighbor, which is indispensable to capture the mechanics of complex systems, such as carbon macromolecules.

In order to provide an easy-to-visualize mechanical picture, the perspective we here adopt is not the one of MD, we consider instead the approach of Favata et al. [1], where a discrete mechanical model is detailed for 2D carbon allotropes. In this view, the configuration of a molecular aggregate is not identified by the coordinates of the mass points, but rather by a suitable finite list of order parameters. In particular, the conditions of *natural equilibrium* of the aggregate can be determined and expressed in terms of such list and independently of the choice of the REBO potential. As we will see, the prediction of such equations is in total agreement with First Principles, Density Functional Theory and Monte Carlo simulations; moreover, given their generality, they can be exploited to establish benchmark solutions.

In order to understand the physical meaning of the conditions we propose, we summarize some of the results of Favata et al. [1]. We make reference to Fig. 1, which depicts a FGS before being rolled up into an achiral CNT. Let the axes 1 and 2 be respectively




Fig. 1. Order parameters in a graphene sheet.

aligned with the armchair and zigzag directions, and let n_1, n_2 be the number of hexagonal cells counted along these axes. On identifying a CNT by its chiral numbers (n, m) , armchair CNTs have $m = n$ and are rolled up from a FGS with $n_1 = 2n$ and n_2 very large; zigzag CNTs have $m = 0$ and are rolled up from a FGS with $n_2 = n$ and n_1 very large. Let us consider now the representative hexagonal cell $A_1B_1A_2B_3A_3B_2A_1$, with sides A_1B_1 and A_3B_3 aligned with the axis 1; the common length of corresponding bonds will be denoted by a , and we will call *a-type* the corresponding bonds. We see that the other four sides have equal length b (*b-type bonds*). We pass to introduce the *bond angles* and, since we intend to consider interactions up to the third neighbor, the *dihedral angles*. As to the bond angles, we notice that they can be of α -type and β -type (e.g., respectively, $\widehat{A_3B_2A_1}$ and $\widehat{B_2A_1B_1}$; see Fig. 1). As to the dihedral angles, there are five types $(\Theta_1, \dots, \Theta_5)$, which can be identified with the help of the colored bond chains in Fig. 1. In conclusion, to determine the deformed configuration of a representative hexagonal cell, no matter if that cell belongs to a FGS or to an achiral CNT, we need to determine the 9-entry *order-parameter substring*:

$$\xi_{sub} := (a, b, \alpha, \beta, \Theta_1, \dots, \Theta_5). \quad (1)$$

The complete order-parameter string for the whole molecular aggregate can be obtained by sequential juxtaposition of substrings. Due to the geometric compatibility conditions induced by the built-in symmetry (see Favata et al. [1] for details), only three of the nine kinematic variables determine the natural configuration, which are chosen to be a, b , and α . In particular, by distinguishing the armchair (superscript *A*) from the zigzag (superscript *Z*) case, the order-parameter substrings are given by, respectively:

$$\xi_{sub}^A = (a, b, \alpha, \tilde{\beta}^A(\alpha, \varphi^A), \tilde{\Theta}_1^A(\alpha, \varphi^A), \tilde{\Theta}_2^A(\alpha, \varphi^A), 2\tilde{\Theta}_2^A(\alpha, \varphi^A), 0, 0); \quad (2)$$

$$\xi_{sub}^Z = (a, b, \alpha, \tilde{\beta}^Z(\alpha, \varphi^Z), \varphi^Z, \tilde{\Theta}_2^Z(\alpha, \varphi^Z), 0, 2\tilde{\Theta}_2^Z(\alpha, \varphi^Z), 0).$$

The explicit form of the functions $\tilde{\beta}^{A,Z}, \tilde{\Theta}_1^A, \tilde{\Theta}_2^{A,Z}$ is given in Favata et al. [1]. In (2), $\varphi^A = \pi/n_1$ is the angle between the plane of $A_1B_1B_3$ and the plane of $B_1A_2B_3$ when an armchair CNT is considered, and $\varphi^Z = \pi/n_2$ the angle between the planes of $A_1B_1A_2$ and $A_2B_3A_3$, when a zigzag CNT is considered. In case of a FGS, we have $\varphi^{A,Z} = 0$, $\tilde{\beta}^{A,Z} = \pi - \alpha/2$, and $\tilde{\Theta}_1^A = \tilde{\Theta}_2^{A,Z} \equiv 0$.

The equilibrium equations turn out to be the following ones:

$$\sigma_a = 0, \quad \sigma_b = 0, \quad \tau_\alpha + 2\beta_{,\alpha} \tau_\beta + \Theta_{1,\alpha} \mathcal{T}_1 + 2\Theta_{2,\alpha} \mathcal{T}_2 + \Theta_{3,\alpha} \mathcal{T}_3 + \Theta_{4,\alpha} \mathcal{T}_4, \quad (3)$$

where $\sigma_a, \sigma_b, \tau_\alpha, \tau_\beta$, and \mathcal{T}_i , are the so-called *nanostresses*, work-conjugate to changes of, respectively, bond lengths, bond angles, and dihedral angles of each type considered. The form of the third

of (3) depends on which of the two achiral CNTs is dealt with; more precisely, we have that

$$\begin{aligned}\tau_\alpha^A + 2\beta^A \tau_\beta^A + \Theta_{1,\alpha}^A \mathcal{T}_1^A + 2\Theta_{2,\alpha}^A \mathcal{T}_2^A + \Theta_{3,\alpha}^A \mathcal{T}_3^A &= 0, \\ \tau_\alpha^Z + 2\beta^Z \tau_\beta^Z + 2\Theta_{2,\alpha}^Z \mathcal{T}_2^Z + \Theta_{4,\alpha}^Z \mathcal{T}_4^Z &= 0.\end{aligned}\quad (4)$$

Due to their generality, the conditions (3) may serve as a benchmark for any REBO potential. To express the equilibrium equations in terms of the Lagrangian coordinates a , b , and α , it is necessary to introduce the constitutive equations for the stress, which result from the assignment of an intermolecular potential. In the next section, we detail the formulas in the Brenner 2nd generation REBO potential [2] which are needed to solve (3) in terms of the order parameters.

2.1. The traction problem

Starting from the geometry and the energy gathered by means of (3), it is possible to obtain secondary quantities. The Young modulus can be computed by solving the equilibrium problem in the presence of a traction load F , whose corresponding governing equations are the following:

$$\sigma_a = 0, \quad (5)$$

$$\sigma_b - \frac{F}{n_1} \sin \frac{\alpha}{2} = 0, \quad (6)$$

$$\begin{aligned}\tau_\alpha + 2\tau_\beta \beta \cos \alpha + \Theta_{1,\alpha} \mathcal{T}_1 + 2\Theta_{2,\alpha} \mathcal{T}_2 \\ + \Theta_{3,\alpha} \mathcal{T}_3 - \frac{1}{2} \frac{F}{n_1} b \cos \frac{\alpha}{2} = 0,\end{aligned}\quad (7)$$

for the armchair traction direction and

$$\sigma_a - \frac{F}{n_2} = 0, \quad (8)$$

$$\sigma_b + \frac{F}{2n_2} \cos \beta = 0, \quad (9)$$

$$\tau_\alpha + 2\tau_\beta \beta \cos \alpha + 2\Theta_{2,\alpha} \mathcal{T}_2 + \Theta_{4,\alpha} \mathcal{T}_4 - \frac{1}{2} \frac{F}{n_2} b \beta \sin \beta = 0, \quad (10)$$

for the zigzag direction. Once these equations have been solved, with the constitutive equations (17), the axial deformation can be computed as:

$$F \mapsto \varepsilon(F) := \frac{\lambda(F) - \lambda_0}{\lambda_0}, \quad (11)$$

where $\lambda(F)$ is the deformed length of the CNT due to the load F and the λ_0 the initial length. The Young modulus is defined to be

$$E(F) = \frac{F}{\varepsilon(F)} \frac{1}{2\pi\rho(F)t}, \quad (12)$$

where $\rho(F)$ is the deformed radius of the CNT after the deformation consequent to the load F and t is the nominal thickness. The evaluation of this latter value is still object of debate, giving rise to the so-called Jakobson's paradox [3]; valuable contributions on the subject are Huang et al. [4], Pine et al. [5] Bajaj et al. [6] and references cited therein. An accurate account of this issue is out of the scope of this paper. Be that as it may, the thickness value does not affect the significance of the present work; in order to compare results from our benchmark with those obtained in LAMMPS, we set $t = 0.34$ nm, a value commonly used by several authors.

For $F \rightarrow 0$, the Young modulus in a neighborhood of the natural configuration is computed. As to the Poisson coefficient, we define it as

$$\nu(F) = -\frac{\rho(F) - \rho_0}{\rho_0} \frac{1}{\varepsilon(F)}, \quad (13)$$

where ρ_0 is the radius in the natural configuration. For $F \rightarrow 0$, its value in a neighborhood of the natural configuration is determined.

3. REBO potentials

In the Brenner 2nd generation REBO potential, the binding energy V^{REBO} of a molecular aggregate is written as a sum over nearest neighbors:

$$V^{\text{REBO}} = \sum_i \sum_{J<I} V_{IJ}; \quad (14)$$

the interatomic potential V_{IJ} is given by the construct

$$V_{IJ} = V_R(r_{IJ}) + b_{IJ} V_A(r_{IJ}), \quad (15)$$

where the individual effects of the *repulsion* and *attraction functions* $V_R(r_{IJ})$ and $V_A(r_{IJ})$, which model pair-wise interactions of atoms I and J depending on their distance r_{IJ} , are modulated by the *bond-order function* b_{IJ} , which depends on the bond angles θ_{IJK} between bonds IJ and JK and on the dihedral angle Θ_{IJKL} between the planes of I, J, K and I, J, L .

When the point of view described in Section 2 is assumed, the expressions of the potentials have to be specialized and written in terms of the order parameters in the substrings (1). On introducing the potentials V_a and V_b for the a - and b -type bonds, we have, respectively:

$$\begin{aligned}V_a(a, \beta, \Theta_1) &= V_R(a) + b_a(\beta, \Theta_1) V_A(a), \\ V_b(b, \alpha, \beta, \Theta_2, \Theta_3, \Theta_4) &= V_R(b) + b_b(\alpha, \beta, \Theta_2, \Theta_3, \Theta_4) V_A(b)\end{aligned}\quad (16)$$

(see Favata et al. [1] for details).

Once this has been done, the nanostresses entering the balance equations (3) can be expressed in terms of the order parameters by means of the following constitutive relations:

$$\begin{aligned}\sigma_a &= V'_R(a) + b_a(\beta, \Theta_1) V'_A(a), \\ \sigma_b &= V'_R(b) + b_b(\alpha, \beta, \Theta_2, \Theta_3, \Theta_4) V'_A(b), \\ \tau_\alpha &= b_{b,\alpha}(\alpha, \beta, \Theta_2, \Theta_3, \Theta_4) V_A(b), \\ \tau_\beta &= \frac{1}{4} (b_{a,\beta}(\beta, \Theta_1) V_A(a) + 2b_{b,\beta}(\alpha, \beta, \Theta_2, \Theta_3, \Theta_4) V_A(b)), \\ \mathcal{T}_1 &= \frac{1}{2} b_{a,\Theta_1}(\beta, \Theta_1) V_A(a), \\ \mathcal{T}_2 &= \frac{1}{2} b_{b,\Theta_2}(\alpha, \beta, \Theta_2, \Theta_3, \Theta_4) V_A(b), \\ \mathcal{T}_3 &= b_{b,\Theta_3}(\alpha, \beta, \Theta_2, \Theta_3, \Theta_4) V_A(b), \\ \mathcal{T}_4 &= b_{b,\Theta_4}(\alpha, \beta, \Theta_2, \Theta_3, \Theta_4) V_A(b).\end{aligned}\quad (17)$$

4. Mathematica program vs LAMMPS results

The most direct outcomes of our solution are natural geometry and energy, which can be used to check the correctness of whatever MD code. The analytical model described has been coded in a *Mathematica* program, that computes the natural radius and the cohesive energy of armchair and zigzag CNTs. The program implements the 2nd generation Brenner potential, but other or customized REBO potentials can be assigned by the user by changing the functions V_R , V_A , b_a , and b_b appearing in (16). Possible alternatives to the Brenner 2nd generation potential are the Tersoff potential [7,8] or the Brenner 1st generation potential [9], which are also readily available in LAMMPS. It is worth noticing that a benchmark for density functional-based codes (such as DFTB, see Aradi et al. [10]), which could serve as alternative methods of computation when samples are not too large, would be much harder to formulate and implement. The results obtained with the program are in good agreement with First Principles, Density Functional Theory (DFT) and Diffusion Monte Carlo (DMC) simulations, as Tables 1 and 2 show. A related point to consider is that our evaluation of the radii is different from that obtained

Table 1
Cohesive energy (eV/atom).

Our benchmark	El-Barbery et al. [12] (First Principles)	Shin et al. 2014 [13] (DMC)
−7.3951	−7.4	−7.464

by classical Roll-Up Model (RUM), which adopts bond lengths shorter in CNTs than in their parent flat graphene sheets, due to the difference between the length of a helix segment and the distance between its endpoints. In an elegant study initiated by Cox and Hill, see Lee et al. [11] and the references cited therein, the geometrical approximation of RUM has been overcome, and precise analytical expressions for the radius have been proposed, in terms of the bond lengths and bond angles. We verified that on inserting our values of bond lengths and bond angles in those formulas, the resulting values for the radius are equal to ours, up to the fourth significant digit, for all considered CNTs.

As an application of the possibility of exploiting the benchmark solutions, we present in Table 3 the results for a number of CNTs, showing that standard LAMMPS code underestimates the geometry and highly overestimates the energy. The origin of the discrepancies can be found only by a close inspection of LAMMPS source code. In fact, although in Brenner et al. [2] it is indicated that the values of the function P_{ij} should be taken null for solid-state carbon, the code assigns the value 0.027603. This latter value is actually dictated in Table VIII of Stuart et al. [18] for AIREBO potentials, due to the additional terms included in this potential. Whenever a LAMMPS user wants to adopt REBO potentials, he needs to change the hard-wired number for the variable PCCF2[0] in “pair_airebo.cpp”; unfortunately, the LAMMPS manual does not provide any information on this issue, and most studies based on LAMMPS REBO calculations are likely to have underestimation or overestimation of mechanical and geometrical properties presented in our Tables. An example of the use of LAMMPS with 2nd generation Brenner potential is Zhang et al. [19]. When the value assigned in Brenner et al. [2] is implemented, the LAMMPS code produces the same results as the benchmark solution, letting alone a tiny difference due to numerical effects, as Table 3 undeniably makes evident.¹

Starting from the geometry and the energy gathered by means of (3), it is possible to obtain secondary quantities. Besides the radius and cohesive energy, the *Mathematica* program yields as output the Young’s modulus and the Poisson coefficient of armchair and zigzag CNTs. In Table 4 some results are reported and compared with standard LAMMPS code: the latter overestimates the Young’s modulus and underestimates the Poisson coefficient. Our results are in very good agreement with the literature (see e.g. Agrawal et al. [20]). The differences between our benchmark and the LAMMPS code with modified parameters are ascribable to numerical effects, more accentuated because Young’s modulus and Poisson coefficients are quantities not directly evaluated, but rather derived, and an increment of numerical error is foreseeable.

5. Description of the software structure and the individual software components

A simple program for solving Eqs. (3) has been implemented in *Mathematica*, version 9. The program, entitled *MDBenchmarks*,

is written in two files: the Package *Benchmark_code.m* and the Computable Document Format *Benchmark_solutions.cdf*, which needs the package to be loaded. In the CDF file it is sufficient to choose armchair or zigzag CNTs and assign the chiral number n to get the benchmark solutions for the 2nd generation Brenner potential, set as default potential. Other REBO potentials can be defined in the package file.

The program *Benchmark_code.m* is divided into four chapters:

1. REBO Potentials.

In this chapter the form of the REBO potential to be tested is assigned. In the section “2nd generation Brenner potential”, the default setting for this potential is implemented, according to [2]; in particular, in the subsection “Potential components” the components introduced in (15) are specified. In the section “Analytical discrete model” the definition of the nanostresses (17) is implemented; this definition is independent of the REBO potential one chooses.

2. Armchair CNTs.

In this chapter the equilibrium problem for armchair CNTs is solved. In the section “Generalities” the geometric conditions on the order parameters are established and the nanostresses are computed. In the section “Solution of the equilibrium equations” the solution of the systems (3)₁ and (4)₁ is determined as a function of the applied force F and the chiral number n . In the section “Radius” the natural radius is computed as a function of F and n and then determined for $F = 0$, namely in the natural configuration. In the section “Energy” the natural energy is computed as a function of F and n and then determined for $F = 0$, namely the cohesive energy. In the section “Young’s modulus” the current and the referential lengths of a CNT are determined, and the strain measure is defined, as a function of F and n ; on introducing the nominal thickness, the Young’s modulus is defined as a function of F and n , and then computed for a tiny value of F , up to convergence. In section “Poisson coefficient”, the named material parameter is defined as a function of F and n , and then computed for a tiny value of F , up to convergence.

3. Zigzag CNTs.

This chapter has the same sections as the previous one, but implemented for the zigzag case; the different geometric constraints are properly included.

4. Summary of results.

In this chapter the benchmark solutions are collected for the visualization in the CDF file *Benchmark_solutions.cdf*.

The software package is supplemented by three folders:

1. *Original_and_Modified_REBOPotFiles*, containing the two LAMMPS files for the original and the modified REBO potential, “pair_airebo.cpp”, instrumental to make the comparison of Tables 3 and 4.
2. *CNT_Graphene_DATAFiles*, containing LAMMPS input files with the coordinates of nanotubes and graphene we examined. These coordinates are obtained by simply mapping atomic locations in graphene to a cylinder. These files can be converted into input files for any other molecular dynamics package.
3. *CNT_Graphene_OUTPUTFiles*, containing files with the coordinates of nanotubes and graphene resulting from the energy minimization in LAMMPS using the modified REBO potential.

Acknowledgments

N.M.P. is supported by the European Research Council (ERC StG Ideas 2011 BIHSNAM No. 279985, ERC PoC 2015 SILKENE No. 693670) and by the European Commission under the Graphene Flagship (WP14 ‘Polymer Composites’, No. 696656). S.R. is supported by the Technology Development Program to solve Climate Changes of the National Research Foundation (NRF) No.

¹ Oftentimes, numerical instabilities (such as ‘exploding’ results) have been reported in internet forums dedicated to LAMMPS simulations. These can be caused by a variety of reasons, not necessarily related to a problem in the potential. At present, we are not aware of any example where the issue we found in the potential causes numerical instabilities.

Table 2
Radii (nm) of small CNTs.

(n, m)	Our benchmark	Machón et al. 2002 [14] (DFT)	Cabria et al. 2003 [15] (DFT)	Popov 2004 [16] (TB)	Budyka et al. 2005 [17] (DFT)
(3, 3)	0.211	0.210	0.212	0.212	–
(4, 4)	0.277	–	–	–	0.277
(5, 0)	0.208	0.204	0.206	0.205	–

Table 3
Geometry and energy.

(n, m)	Radii (nm)			Cohesive energy (ev/atom)		
	Our benchmark	LAMMPS (standard)	LAMMPS (modified)	Our benchmark	LAMMPS (standard)	LAMMPS (modified)
(3, 3)	0.2111	0.2079	0.2110	–7.0137	–7.3838	–7.0137
(4, 4)	0.2767	0.2723	0.2766	–7.1695	–7.5569	–7.1695
(5, 5)	0.3431	0.3371	0.3404	–7.2463	–7.6422	–7.2462
(6, 6)	0.4101	0.4035	0.4100	–7.2898	–7.6905	–7.2896
(7, 7)	0.4773	0.4697	0.4773	–7.3167	–7.7204	–7.3166
(8, 8)	0.5447	0.5361	0.5447	–7.3346	–7.7403	–7.3345
(10, 10)	0.6798	0.6690	0.6798	–7.3560	–7.7640	–7.3558
(12, 12)	0.8151	0.8022	0.8151	–7.3678	–7.7771	–7.3676
(18, 18)	1.2216	1.2020	1.2215	–7.3829	–7.7038	–7.3827
(25, 25)	1.6961	1.6689	1.6960	–7.3887	–7.8003	–7.3886
(5, 0)	0.2076	0.2046	0.2076	–6.9758	–7.3417	–6.9759
(6, 0)	0.2446	0.2409	0.2446	–7.0969	–7.4763	–7.0969
(7, 0)	0.2821	0.2778	0.2821	–7.1716	–7.5593	–7.1715
(8, 0)	0.3201	0.3151	0.3201	–7.2212	–7.6144	–7.2212
(9, 0)	0.3583	0.3527	0.3583	–7.2561	–7.6531	–7.2560
(10, 0)	0.3967	0.3905	0.3967	–7.2814	–7.6812	–7.2813
(12, 0)	0.4739	0.4665	0.4739	–7.3151	–7.7186	–7.3149
(15, 0)	0.5904	0.5810	0.5904	–7.3432	–7.7499	–7.3431
(20, 0)	0.7853	0.7274	0.7852	–7.3656	–7.7747	–7.3655
(30, 0)	1.1760	1.1572	1.1759	–7.3819	–7.7927	–7.3818
Graphene				–7.3951	–7.8074	–7.3950

Table 4
Material properties.

(n, m)	Young's modulus (GPa)			Poisson coefficient		
	Our benchmark	LAMMPS (standard)	LAMMPS (modified)	Our benchmark	LAMMPS (standard)	LAMMPS (modified)
(3, 3)	890.5525	987.0102	885.0631	0.1490	0.1237	0.1563
(4, 4)	848.7784	944.4810	840.1351	0.2343	0.2078	0.2388
(5, 5)	793.9856	901.869	799.7630	0.2997	0.2578	0.2963
(6, 6)	817.0959	891.6427	789.8102	0.3012	0.2794	0.3115
(7, 7)	779.4158	881.2895	778.2888	0.3202	0.2937	0.3318
(8, 8)	766.6752	872.9931	767.9165	0.3325	0.3020	0.3458
(10, 10)	754.2882	856.9461	756.7625	0.3629	0.3194	0.3526
(12, 12)	755.9326	846.2911	746.5046	0.3458	0.3306	0.3626
(18, 18)	740.6592	831.2163	732.5252	0.3816	0.3426	0.3740
(25, 25)	733.2746	823.3865	726.4968	0.3808	0.3507	0.3790
(5, 0)	947.9966	1046.3569	943.9120	0.0655	0.0362	0.0661
(6, 0)	972.9604	1075.4912	968.5679	0.0868	0.0600	0.0840
(7, 0)	976.4790	1082.0265	971.5102	0.1100	0.0800	0.1105
(8, 0)	969.7586	1076.6910	965.8188	0.1328	0.1045	0.1307
(9, 0)	958.1857	1066.6410	954.9396	0.1544	0.1234	0.1511
(10, 0)	944.4616	1053.1830	941.3135	0.1743	0.1424	0.1737
(12, 0)	916.1866	1025.8253	915.1339	0.2086	0.1725	0.2032
(15, 0)	877.4224	986.5745	877.9641	0.2493	0.2138	0.2447
(20, 0)	830.2633	940.5973	835.7993	0.2949	0.2564	0.2835
(30, 0)	779.997	890.5147	789.0504	0.3405	0.2991	0.3461

2015M1A2A2056561, funded by the Ministry of Science, ICT & Future Planning. The authors would like to thank Prof. Paolo Podio-Guidugli for fruitful discussions.

References

- [1] A. Favata, A. Micheletti, P. Podio-Guidugli, N. Pugno, J. Elasticity (2016) <http://dx.doi.org/10.1007/s10659-015-9568-8>.
- [2] D. Brenner, O. Shenderova, J. Harrison, S. Stuart, B. Ni, S. Sinnott, J. Phys.: Condens. Matter 14 (4) (2002) 783.
- [3] O.A. Shenderova, V.V. Zhirnov, D.W. Brenner, Crit. Rev. Sol. State 27 (3–4) (2002) 227–356.
- [4] Y. Huang, J. Wu, K.C. Hwang, Phys. Rev. B 74 (2006) 245413.
- [5] P. Pine, Y. Yaish, J. Adler, Phys. Rev. B 83 (2011) 155410.
- [6] C. Bajaj, A. Favata, P. Podio-Guidugli, Eur. J. Mech. A Solids 42 (2013) 137–157.
- [7] J. Tersoff, Phys. Rev. B 37 (1988).
- [8] J. Tersoff, Phys. Rev. B 39 (1989).
- [9] D. Brenner, Phys. Rev. B 42 (15) (1990).
- [10] B. Aradi, B. Hourahine, T. Frauenheim, J. Phys. Chem. A 111 (26) (2007) 5678–5684.
- [11] R.F. Lee, B. Cox, J. Hill, Nanoscale 2 (2010) 859–872.

- [12] A. El-Barbary, R. Telling, C. Ewels, M. Heggie, P. Briddon, *Phys. Rev. B* 68 (2003) 144107.
- [13] H. Shin, S. Kang, J. Koo, H. Lee, J. Kim, Y. Kwon, J. Chem. Phys. 140 (11) (2014).
- [14] M. Machón, S. Reich, C. Thomsen, D. Sánchez-Portal, P. Ordejón, *Phys. Rev. B* 66 (2002).
- [15] I. Cabria, J.W. Mintmire, C.T. White, *Phys. Rev. B* 67 (2003) 121406.
- [16] V.N. Popov, *New J. Phys.* 6 (2004) 17.
- [17] M.F. Budyka, T.S. Zyubina, A.G. Ryabenko, S.H. Lin, A.M. Mebel, *Chem. Phys. Lett.* 407 (2005) 266–271.
- [18] S.J. Stuart, A.B. Tutein, J.A. Harrison, *J. Chem. Phys.* 112 (2000) 6472.
- [19] P. Zhang, L. Ma, F. Fan, Z. Zeng, C. Peng, P. Loya, Z. Liu, Y. Gong, J. Zhang, X. Zhang, P. Ajayan, T. Zhu, J. Lou, *Nature Commun.* 5 (2014).
- [20] P. Agrawal, B. Sudalayandi, L. Raff, R. Komanduri, *Comp. Mat. Sci.* 38 (2) (2006) 271–281.

```
Clear["Global`*"]
```

MD Benchmark Code

The program is divided in four chapters:

1. REBO Potentials.

In this chapter the form of the REBO potential to be tested is assigned. In the section "2nd generation Brenner potential", the default setting for this potential is implemented, according to [2]; in particular, in the subsection "Potential components" the components introduced in (17) are specified. In the section "Analytical discrete model" the definition of the nanostresses (19) is implemented; this definition is independent of the REBO potential one chooses.

2. Armchair CNTs

In this chapter the equilibrium problem for armchair CNTs is solved. In the section "Generalities" the geometric conditions on the order parameters are established and the nanostresses are computed. In the section "Solution of the equilibrium equations" the solution of the system 3_1 and 4_1 is determined as a function of the applied force F and the chiral number n . In the section "Radius" the natural radius is computed as a function of F and n and then determined for $F=0$, namely in the natural configuration. In the section "Energy" the natural energy is computed as a function of F and n and then determined for $F=0$, namely the cohesive energy. In the section "Young's modulus" the current and the referential lengths of a CNT are determined, and the strain measure is defined, as a function of F and n ; on introducing the nominal thickness, the Young's modulus is defined as a function of F and n , and then computed for a tiny value of F , up to convergence. In section "Poisson coefficient", the named material parameter is defined as a function of F and n , and then computed for a tiny value of F , up to convergence.

3. Zigzag CNTs.

This chapter has the same section as the previous one, but implemented for the zigzag case; the different geometric constraints are properly included.

4. Summary of Results.

In this chapter the benchmark solutions are collected for the visualization in the CDF file `Benchmark_solutions.cdf`.

Equation numbers and bibliographic reference numbers are those of the article associated to this code.

I. REBO Potentials

2nd generation Brenner potential

G Function

According to the definition in reference [2]

```

d1={Subscript[d, 0]-> 0.375449999999999,Subscript[d, 1]->1.406776475152104,
Subscript[d, 2]->2.254377494462425,Subscript[d, 3]->2.031282890266629,
Subscript[d, 4]->1.429711740681566,Subscript[d, 5]->0.502401399437276};
G1[θ_]=Sum[d_i*Cos[θ]^i/.d1;

d2={Subscript[d, 0]->0.707277245734129,Subscript[d, 1]->5.677435848489921,
Subscript[d, 2]->24.09701877750397,Subscript[d, 3]->57.59183195999613,
Subscript[d, 4]->71.88287000288395,Subscript[d, 5]->36.27886067346856};
G2[θ_]=Sum[d_i*Cos[θ]^i/.d2;

d3={Subscript[d, 0]->-0.6443999999999948,Subscript[d, 1]->-6.207999999999618,
Subscript[d, 2]->-20.05899999999891,Subscript[d, 3]->-30.22799999999852,
Subscript[d, 4]->-21.72399999999905,Subscript[d, 5]->-5.990399999999763};
G3[θ_]=Sum[d_i*Cos[θ]^i/.d3;
G[θ_]=Piecewise[{{G1[θ],0<=θ<0.6082*π},{G2[θ],0.6082*π<=θ<2/3*π},{G3[θ],2/3*π<=θ<=π}}];

```

Potential parameters

```
ev=0.1602176487;(*unit conversion constant*)
```

```

potparameters={Subscript[B, 1]-> 12388.79197798,Subscript[B, 2]->17.56740646509,
Subscript[B, 3]->30.71493208065, Subscript[β, 1]->4.7204523127,Subscript[β, 2]->1.4332132499,
Subscript[β, 3]->1.3826912506,Q->0.3134602960833,A->10953.544162170,α0->4.7465390606595};

```

```
T=-0.004048375;
```

Potential components

equation (17)
(reference [2])

```

VA[r_]=-Sum[B_n*E^-β_n*r/.potparameters;
VR[r_]=(1+Q/r)*A*E^(-α0*r)/.potparameters;
DVA[r_]=D[VA[r],r];
DVR[r_]=D[VR[r],r];
ba=(1+2*G[β])^(-(1/2))+2*T*(1-Cos[θ1]^2);
bb=(1+G[α]+G[β])^(-(1/2))+T*(2*(1-Cos[θ2]^2)+(1-Cos[θ3]^2)+(1-Cos[θ4]^2));

```

Analytical discrete model

Geometry

```

n1[n_]:=2*n;
n2[n_]:=n;

```

Nanostresses

Equation (19)

```

σa=DVR[a]+ba*DVA[a];
σb=DVR[b]+bb*DVA[b];
τα=D[bb,α]*VA[b];
τβ=1/4*(D[ba,β]*VA[a]+2*D[bb,β]*VA[b]);
T1=1/2*D[ba,θ1]*VA[a];
T2=1/2*D[bb,θ2]*VA[b];
T3=D[bb,θ3]*VA[b];
T4=D[bb,θ4]*VA[b];

```

2. Armchair CNTs

Generalities

Geometric constraints

Reference [1]

```
φ[n_]:=π/n1[n];
```

```

β[n_]:=ArcCos[-Cos[α/2]*Cos[φ[n]]];
θ1[n_]:=2*ArcSin[(Cos[α/2]*Sin[φ[n]])/Sin[β[n]]];
θ2[n_]:=ArcSin[Sin[φ[n]]/Sin[β[n]]];
θ3[n_]:=2*θ2[n];
θ4[n_]:=0;

```

```

σan[n_]:=σa/.{β->β[n],θ1->θ1[n],θ2->θ2[n],θ3->θ3[n],θ4->θ4[n]}
σbn[n_]:=σb/.{β->β[n],θ1->θ1[n],θ2->θ2[n],θ3->θ3[n],θ4->θ4[n]}
ταn[n_]:=τα/.{β->β[n],θ1->θ1[n],θ2->θ2[n],θ3->θ3[n],θ4->θ4[n]}
τβn[n_]:=τβ/.{β->β[n],θ1->θ1[n],θ2->θ2[n],θ3->θ3[n],θ4->θ4[n]}
T1n[n_]:=T1/.{β->β[n],θ1->θ1[n],θ2->θ2[n],θ3->θ3[n],θ4->θ4[n]}
T2n[n_]:=T2/.{β->β[n],θ1->θ1[n],θ2->θ2[n],θ3->θ3[n],θ4->θ4[n]}
T3n[n_]:=T3/.{β->β[n],θ1->θ1[n],θ2->θ2[n],θ3->θ3[n],θ4->θ4[n]}
T4n[n_]:=T4/.{β->β[n],θ1->θ1[n],θ2->θ2[n],θ3->θ3[n],θ4->θ4[n]}

```

Solution of the equilibrium equations

```

solA[F_,n_]:=FindRoot[{σan[n]==0,σbn[n]-F/n1[n]*Sin[α/2]==0,ταn[n]+
2*D[β[n],α]*τβn[n]+D[θ1[n],α]*T1n[n]+2*D[θ2[n],α]*T2n[n]+D[θ3[n],α]*T3n[n]-
1/2*F/n1[n]*b*Cos[α/2]==0},{a,1.4},{b,1.4},{α,2}]

```

Radius

Reference [1]

```

ρa[n_]:= (b/2*Cos[α/2]+a/2*Cos[φ[n]])/Sin[φ[n]];
ρ[n_]:=Sqrt[ρa[n]^2+a^2/4];
rA[F_,n_]:=ρ[n]/10/.*solA[F,n];(*radius in nm*)

```

Natural radius (output in Table 3)

```
 $\rho A[n_] := rA[0, n]$ 
```

Energy

```
Va=VR[a]+ba*VA[a];  
Vb=VR[b]+bb*VA[b];
```

```
V[F_, n_] := (Va+2*Vb)/2 /. { $\beta \rightarrow \beta[n]$ ,  $\theta 1 \rightarrow \theta 1[n]$ ,  $\theta 2 \rightarrow \theta 2[n]$ ,  $\theta 3 \rightarrow \theta 3[n]$ ,  $\theta 4 \rightarrow \theta 4[n]$ } /. solA[F, n]
```

Cohesive energy (output of Table 3)

Reference [1]

```
EnergyA[n_] := V[0, n] (*ev/atom*)
```

Young Modulus

Equations (13)-(14)

```
 $\lambda A[n_] := 2 * \text{Sin}[\alpha/2] * n2[n] * b$ ; (*current length*)  
 $\lambda 0A[n_] := \lambda A[n] /. \text{solA}[0, n]$ ; (*referential length*)  
 $\epsilon A[n_] := (\lambda A[n] - \lambda 0A[n]) / \lambda 0A[n]$ ; (*strain measure*)  
 $eA[F_, n_] := \epsilon A[n] /. \text{solA}[F, n]$ ;
```

```
t=0.34(*nm*); (*nominal thickness*)
```

```
YA[F_, n_] := (F/eA[F, n]) * ev * (1/(2 *  $\pi$  * rA[F, n] * t)) * 10 (*Young modulus, GPa*)
```

Young modulus in the origin (output in Table 4)

```
YoungA[n_] := If[n < 20, YA[10-10, n], YA[10-9, n]] (*GPa*)
```

Poisson coefficient

Equation (15)

```
 $\nu A[F_, n_] := -((rA[F, n] - rA[0, n]) / rA[0, n]) * 1 / eA[F, n]$ 
```

Poisson coefficient in the origin (output in Table 4)

```
PoissonA[n_] := If[n < 20,  $\nu A[10^{-10}, n]$ ,  $\nu A[10^{-9}, n]$ ]
```

3. Zigzag CNTs

Generalities

Geometric constraints

Reference [1]

```
 $\phi z[n\_]:= \pi/n;$ 
```

```
 $\beta z[n\_]:= \pi - \text{ArcSin}[\text{Sin}[\alpha/2]/\text{Cos}[\phi z[n]/2]];$   
 $\theta 1z[n\_]:= \phi z[n];$   
 $\theta 2z[n\_]:= \text{ArcSin}[(\text{Sin}[\beta z[n]]*\text{Sin}[\phi z[n]])/\text{Sin}[\alpha]];$   
 $\theta 3z[n\_]:= 0;$   
 $\theta 4z[n\_]:= 2*\theta 2z[n];$ 
```

```
 $\sigma a z[n\_]:= \sigma a /. \{\beta \rightarrow \beta z[n], \theta 1 \rightarrow \theta 1z[n], \theta 2 \rightarrow \theta 2z[n], \theta 3 \rightarrow \theta 3z[n], \theta 4 \rightarrow \theta 4z[n]\}$   
 $\sigma b z[n\_]:= \sigma b /. \{\beta \rightarrow \beta z[n], \theta 1 \rightarrow \theta 1z[n], \theta 2 \rightarrow \theta 2z[n], \theta 3 \rightarrow \theta 3z[n], \theta 4 \rightarrow \theta 4z[n]\}$   
 $\tau \alpha z[n\_]:= \tau \alpha /. \{\beta \rightarrow \beta z[n], \theta 1 \rightarrow \theta 1z[n], \theta 2 \rightarrow \theta 2z[n], \theta 3 \rightarrow \theta 3z[n], \theta 4 \rightarrow \theta 4z[n]\}$   
 $\tau \beta z[n\_]:= \tau \beta /. \{\beta \rightarrow \beta z[n], \theta 1 \rightarrow \theta 1z[n], \theta 2 \rightarrow \theta 2z[n], \theta 3 \rightarrow \theta 3z[n], \theta 4 \rightarrow \theta 4z[n]\}$   
 $T1z[n\_]:= T1 /. \{\beta \rightarrow \beta z[n], \theta 1 \rightarrow \theta 1z[n], \theta 2 \rightarrow \theta 2z[n], \theta 3 \rightarrow \theta 3z[n], \theta 4 \rightarrow \theta 4z[n]\}$   
 $T2z[n\_]:= T2 /. \{\beta \rightarrow \beta z[n], \theta 1 \rightarrow \theta 1z[n], \theta 2 \rightarrow \theta 2z[n], \theta 3 \rightarrow \theta 3z[n], \theta 4 \rightarrow \theta 4z[n]\}$   
 $T3z[n\_]:= T3 /. \{\beta \rightarrow \beta z[n], \theta 1 \rightarrow \theta 1z[n], \theta 2 \rightarrow \theta 2z[n], \theta 3 \rightarrow \theta 3z[n], \theta 4 \rightarrow \theta 4z[n]\}$   
 $T4z[n\_]:= T4 /. \{\beta \rightarrow \beta z[n], \theta 1 \rightarrow \theta 1z[n], \theta 2 \rightarrow \theta 2z[n], \theta 3 \rightarrow \theta 3z[n], \theta 4 \rightarrow \theta 4z[n]\}$ 
```

Solution of the equilibrium equations

```
 $\text{solZ}[F_, n_] := \text{FindRoot}[\{\sigma a z[n] - F/n2[n] == 0, \sigma b z[n] + F/(2*n2[n]) * \text{Cos}[\beta z[n]] == 0,$   
 $\tau \alpha z[n] + 2*D[\beta z[n], \alpha] * \tau \beta z[n] + 2*D[\theta 2z[n], \alpha] * T2z[n] + D[\theta 4z[n], \alpha] * T4z[n]$   
 $- 1/2 * F/n2[n] * b * D[\beta z[n], \alpha] * \text{Sin}[\beta z[n]] == 0\}, \{\{a, 1.4\}, \{b, 1.4\}, \{\alpha, 2\}\}]$ 
```

Radius

Reference [1]

```
 $r z[n_] := \text{Sin}[\beta z[n]] / (2 * \text{Sin}[\phi z[n]/2]) * b;$   
 $r[F_, n_] := r z[n] / 10 /. \text{solZ}[F, n]; (*radius in nm*)$ 
```

Natural radius (output in Table 3)

```
 $\rho Z[n_] := r[0, n] (*nm*)$ 
```

Energy

```
 $VZ[F_, n_] := (Va + 2*Vb) / 2 /. \{\beta \rightarrow \beta z[n], \theta 1 \rightarrow \theta 1z[n], \theta 2 \rightarrow \theta 2z[n], \theta 3 \rightarrow \theta 3z[n], \theta 4 \rightarrow \theta 4z[n]\} /. \text{solZ}[F, n]$ 
```

Cohesive energy (output in Table 3)

```
 $\text{EnergyZ}[n_] := VZ[0, n] (*ev/atom*)$ 
```

Young Modulus

Equations (13)-(14)

```

λZ[n_] := (1 - b/a * Cos[βz[n]]) * n1[n] * a; (*current length*)
λ0Z[n_] := λZ[n] /. solZ[0, n]; (*referential length*)
εZ[n_] := (λZ[n] - λ0Z[n]) / λ0Z[n]; (*strain measure*)
eZ[F_, n_] := εZ[n] /. solZ[F, n];

YZ[F_, n_] := (F / eZ[F, n]) * ev * (1 / (2 * π * r[F, n] * t)) * 10 (*Young modulus, GPa*)

```

Young modulus in the origin (output in Table 4)

```
YoungZ[n_] := YZ[10^-8, n] (*GPa*)
```

Poisson ratio

Equation (15)

```
v[F_, n_] := -((r[F, n] - r[0, n]) / r[0, n]) * 1 / eZ[F, n]
```

Poisson coefficient in the origin (output in Table 4)

```
PoissonZ[n_] := v[10^-8, n]
```

4. Summary of Results

```

Radius[0, n_] := ρA[n]
Radius[1, n_] := ρZ[n]
Energy[0, n_] := EnergyA[n]
Energy[1, n_] := EnergyZ[n]
Young[0, n_] := YoungA[n]
Young[1, n_] := YoungZ[n]
Poisson[0, n_] := PoissonA[n]
Poisson[1, n_] := PoissonZ[n]

```

Creation of the shell for cdf file

```

Manipulate[
Column[{ NumberForm[Radius[q, n], 4] "radius [nm]:", NumberForm[Energy[q, n], 5] "cohesive energy",
NumberForm[Young[q, n], 7] "Young modulus [GPa]:", NumberForm[Poisson[q, n], 4] "Poisson coefficient:"},
Left, " ",
{{q, 0, "show"}, {0 -> "armchair", 1 -> "zigzag"}, Setter},
{n, 3, 30},
AutorunSequencing -> {1, 2},
TrackedSymbols -> {q, n},
ControlPlacement -> Top,
SaveDefinitions -> True,
ControlType -> InputField]

```

Benchmark solutions for REBO potentials

Get ["/mydirectory/Benchmark_code.m"]

show

n

```
radius [nm]: 0.2111
cohesive energy [ev/atom]: -7.0137
Young modulus [GPa]: 890.5525
Poisson coefficient: 0.149
```

Instructions

1. Load the package "Benchmark_code.m" inserting the directory in "Get" command and press "Shift+Enter"
2. Choose armchair or zigzag CNTs
3. Choose the chiral number n
4. Press Enter to visualize the benchmark results

Note

The code is set for 2nd generation Brenner potential. In order to check another REBO potential, it is necessary to define the proper potential functions and all the corresponding parameters in the "Benchmark_code.m" file.